
	Reason for change:
	The TTCN CR R5s150689 was previously agreed, but the implementation is missing in the week38 TTCN delivery.
From R5s150689:

Currently following switch off on UTRAN, the TTCN implementation of the postamble sequence will allow receiving an RRCConnectionRequest with an establishment cause of “detach” or “originatingBackgroundCall” (added for IMS deregistration by TTCN CR R5s140572).

In the case of e.g. IMS emergency call test case 19.1.3 from 34.229-1, the UE will have had two default EPS bearer contexts and one dedicated EPS bearer context active on EUTRAN, at the point that it is redirected / reselects to UTRAN. This will map to 3 PDP contexts active in UTRAN, two with “background” class (corresponding to the default EPS bearers) and one with “conversational” class (corresponding to the dedicated EPS bearer). On switch off, when the UE wishes to re-establish a PS RAB in order to perform IMS deregistration, it will set the establishment cause in the RRCConnectionRequest according 24.008 Annex L:

If the request is not to re-establish RABs for emergency bearer services, the RRC establishment cause shall be set to either ‘Originating Conversational Call’ or ‘Originating Streaming Call’ or ‘Originating Interactive Call’ or ‘Originating Background Call ‘ – depending on the Traffic Class in QoS of the "most demanding" Traffic Class, considering all active PDP contexts. (see Note 1)

NOTE 1:            For classification of "most demanding" Traffic Class the following ranking order applies: "Conversational" followed by "Streaming" followed by "Interactive" followed by "Background", where "Conversational" is the most demanding Traffic class in terms of being delay sensitive.
Following this, the UE will set the establishment cause to “OriginatingConversationalCall” in this scenario.

	
	

	Summary of change:
	Re-implement the changes from TTCN CR R5s150689.
Allow any establishment cause in RRCConnectionRequest received following switch off on UTRAN

	
	

	Consequences if not approved:
	A conformant UE can fail IMS emergency test case 19.1.3 in postamble sequence.

Change 1
	Function name
	

	Reason for change
	Currently following switch off on UTRAN, the TTCN implementation will allow receiving an RRCConnectionRequest with an establishment cause of “detach” or “originatingBackgroundCall” (added for IMS deregistration by CR R5s140572).

In the case of e.g. IMS emergency call test case 19.1.3 from 34.229-1, the UE will have had two default EPS bearer contexts and one dedicated EPS bearer context at the point that it is redirected to UTRAN. This will map to 3 PDP contexts active in UTRAN, two with “background” class (corresponding to the default EPS bearers) and one with “conversational” class (corresponding to the dedicated EPS bearer). On switch off, when the UE wishes to re-establish a PS RAB in order to perform IMS deregistration, it will set the establishment cause in the RRCConnectionRequest according 24.008 Annex L:

If the request is not to re-establish RABs for emergency bearer services, the RRC establishment cause shall be set to either ‘Originating Conversational Call’ or ‘Originating Streaming Call’ or ‘Originating Interactive Call’ or ‘Originating Background Call ‘ – depending on the Traffic Class in QoS of the "most demanding" Traffic Class, considering all active PDP contexts. (see Note 1)

NOTE 1:            For classification of "most demanding" Traffic Class the following ranking order applies: "Conversational" followed by "Streaming" followed by "Interactive" followed by "Background", where "Conversational" is the most demanding Traffic class in terms of being delay sensitive.
Following this, the UE will set the establishment cause to “OriginatingConversationalCall” in this scenario.

	Summary of change
	Allow any establishment cause in RRCConnectionRequest received following switch off on UTRAN (by not passing any 2nd parameter to function f_UTRAN_RRC_ConnEst()).


	TTCN module
	UTRAN_CommonProcedures.ttcn

	MCC160 Comments
	Accepted.  See MCC160 implementation below


Before change
   function f_UTRAN_SwitchPower_Off(UTRAN_CellId_Type         p_CellId,

                                   UTRA_POSTAMBLE_STATE_Type p_Utra_Postamble_State,

                                   UTRAN_GERAN_ATTACH_Type p_AttachType,

                                   boolean p_USIM_Rmvd := false,

                                   boolean p_UsedStoredNsapi := false) //@sic R5s141127 sic@

    runs on UTRAN_PTC

  {

  ………….

    //Step 1: The UE can be switched off or USIM removed

    if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) { // @sic R5s120597r1 sic@

      //Step 1: The UE is switched off

      f_UT_SwitchOffUE(UT, p_USIM_Rmvd);

      //Step 2-4: RRC Connection establishment when UE is in UTRA idle end state

      if (p_Utra_Postamble_State == U1_IDLE) {

        f_UTRAN_RRC_ConnEst(p_CellId, (detach, originatingBackgroundCall)); // @sic R5s140572 sic@

……………

}

After change
   function f_UTRAN_SwitchPower_Off(UTRAN_CellId_Type         p_CellId,

                                   UTRA_POSTAMBLE_STATE_Type p_Utra_Postamble_State,

                                   UTRAN_GERAN_ATTACH_Type p_AttachType,

                                   boolean p_USIM_Rmvd := false,

                                   boolean p_UsedStoredNsapi := false) //@sic R5s141127 sic@

    runs on UTRAN_PTC

  {

  ………….

    //Step 1: The UE can be switched off or USIM removed

    if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) { // @sic R5s120597r1 sic@

      //Step 1: The UE is switched off

      f_UT_SwitchOffUE(UT, p_USIM_Rmvd);

      //Step 2-4: RRC Connection establishment when UE is in UTRA idle end state

      if (p_Utra_Postamble_State == U1_IDLE) {

        f_UTRAN_RRC_ConnEst(p_CellId); // @sic R5s140572 sic@

……………

}

MCC160 Implementation:
  function f_UTRAN_SwitchPower_Off(UTRAN_CellId_Type         p_CellId,

                                   UTRA_POSTAMBLE_STATE_Type p_Utra_Postamble_State, // @sic R5s150895 sic@

                                   UTRAN_GERAN_ATTACH_Type p_AttachType,

                                   boolean p_USIM_Rmvd := false,

                                   boolean p_UsedStoredNsapi := false) //@sic R5s141127 sic@

    runs on UTRAN_PTC

  {

    var RB_ConfigType v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);

    var RRC_Rel_Status v_RRC_Rel_Status := cell_Dch;

    var UTRA_POSTAMBLE_CONDITIONS v_Utra_Postamble_Condition := f_UTRA_SetPostambleCondition(p_AttachType); // @sic R5s110176 sic@

    var RRC_DATA_IND v_RRC_DataInd;

    var U_RLC_AM_IND v_U_RLC_AM_IND;

    var UTRAN_SecurityInfo_Type v_SecurityInfo ;

    var boolean v_RxdServiceReq := false; // Should only get one

    if (v_CellConfig == cell_FACH or v_CellConfig == cell_FACH_NoConn or v_CellConfig == cell_FACH_PS) { //@sic R5s150302 sic@

      v_RRC_Rel_Status := cell_Fach_Dcch;

    }

    //Step 1: The UE can be switched off or USIM removed

    if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) { // @sic R5s120597r1 sic@

      //Step 1: The UE is switched off

      f_UT_SwitchOffUE(UT, p_USIM_Rmvd);

      //Step 2-4: RRC Connection establishment when UE is in UTRA idle end state

      if (p_Utra_Postamble_State == U1_IDLE) { 

        f_UTRAN_RRC_ConnEst(p_CellId); // @sic R5s140572, R5s150899 sic@

      } // @sic R5s150895 sic@

        alt {

          [not v_RxdServiceReq] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,

                                                    tsc_RB3,

                                                    cr_U_ServiceRequest(cr_ServiceType('001'B),//Service type 'data'

                                                                   cr_MobileIdAny,

                                                                   ?))) -> value v_RRC_DataInd

            {  // @sic R5-142919, R5s140710 sic@

              f_UTRAN_RRC_Security(p_CellId, false, ps_domain);

              v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);  // @sic R5s150057 sic@

              select (v_CellConfig){

                case (cell_FACH) {

                    f_UTRAN_RB_SetUp_PS_RAB (p_CellId, cell_FACH_PS, false, -, p_UsedStoredNsapi); // @sic R5s150057, R5s150919 sic@

                }

                case else {

                    f_UTRAN_RB_SetUp_PS_RAB (p_CellId, cell_DCH_64kPS_RAB_SRB, false, -, p_UsedStoredNsapi); // @sic R5s140710 R5s141127 R5s150919 sic@

                }

              }

              U_AM.receive (car_RLC_Data_Ind (utran_CellDedicated, //@sic R5s110210 sic@

                             tsc_RB2,

                             cr_108_RB_SetUpCmpl (tsc_RRC_TI))) -> value v_U_RLC_AM_IND;

              v_SecurityInfo := f_UTRAN_Security_Get();

              v_SecurityInfo.start_PS := v_U_RLC_AM_IND.data.aM_message.uL_DCCH_Message.message_.radioBearerSetupComplete.start_Value;

              f_UTRAN_Security_Set(v_SecurityInfo);

              v_RxdServiceReq := true;

              repeat; // now need to detach at NAS level

            }

          [] f_UTRAN_UE_DetachAfterSwitchOff_Common (p_CellId, v_Utra_Postamble_Condition); // @sic R5s150895 sic@

         } // End of alt

         if (v_RxdServiceReq) { // @sic R5-142919, R5s150895 sic@

           f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]); // tell the IMS PTC to stop waiting for the optional deregister

         }

      //Step 6-7: Release RRC connection when the UE has been switched off

      f_UTRAN_RRC_ConnRel_AfterSwitchOff(p_CellId, v_RRC_Rel_Status);

    }

    //Step 1: The power must be removed from the UE

    else {

      f_UT_PowerOffUE (UT);

    }

  } // f_UTRAN_SwitchPower_Off
  altstep f_UTRAN_UE_DetachAfterSwitchOff_Common(UTRAN_CellId_Type         p_CellId,

                                              UTRA_POSTAMBLE_CONDITIONS p_Utra_Postamble_Condition) runs on UTRAN_PTC 

  {

    var integer v_AttFlag := f_UTRAN_CellInfo_GetAttFlag(p_CellId);

    var B2_Type v_Nmo := f_UTRAN_CellInfo_GetNMO(p_CellId);

    //Step 5a1, 5c1 & 5d1: Receive Detach Request message for the different scenarios

    // - NMO1 and (C1 or C3) : detach type = power switched off, GPRS/IMSI combined detach

    // - (NMO1 or NMO2) and C2 : detach type = power switched off, PS detach

    // - NMO2 and (C1 or C3) : detach type = power switched off, PS detach

    [] U_Dc.receive((car_PS_InitDirectTransfer(utran_CellDedicated,

                                            tsc_RB3,

                                            cr_U_DetachRequest), // @sic R5s110462 sic@

                      car_PS_UplinkDirectTransfer(utran_CellDedicated, // if received Service Req, then this won't be the first message

                                                 tsc_RB3,

                                                 cr_U_DetachRequest))) // @sic R5s120162 sic@

    {

      //Step 5d2: Receive IMSI Detach indication

      // - NMO2 and (C1 or C3)

      if (v_AttFlag == 1 and

          ((v_Nmo == tsc_NMO_II and (p_Utra_Postamble_Condition == UTRA_C1 or p_Utra_Postamble_Condition == UTRA_C3)))) { // @sic R5s110155, R5-115770 sic@

        U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,  // @sic R5s110155, R5s110695, R5s120529 sic@

                                            tsc_RB3,

                                            cr_U_Imsi_DetachInd));

      }

    }

    // Step 5b1 or 5d2: Receive IMSI detach Indication

    // - (NMO1 or NMO2) and C4

    // When NMO2 and (C1 or C3) : CS and PS are run independently and SS shall accept either PS detach first or IMSI detach first.

    [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,

                                         tsc_RB3,

                                         cr_U_Imsi_DetachInd)) // @sic R5s120902, R5s130273 sic@

    {

      //Step 5d1: Receive Detach request for the scenario

      // - NMO2 and (C1 or C3)

      if (v_AttFlag == 1 and

          v_Nmo == tsc_NMO_II and // @sic R5-115770 sic@

          (p_Utra_Postamble_Condition == UTRA_C1 or p_Utra_Postamble_Condition == UTRA_C3)) {

        U_Dc.receive((car_PS_InitDirectTransfer(utran_CellDedicated,

                                               tsc_RB3,

                                               cr_U_DetachRequest), // @sic R5s110462 sic@

                      car_PS_UplinkDirectTransfer(utran_CellDedicated, // if received Service Req, then this won't be the first message

                                                 tsc_RB3,

                                                 cr_U_DetachRequest))); // @sic R5s110462 sic@

      }

      else {//IMSI Detach only

      }

    }

  } // f_UTRAN_UE_DetachAfterSwitchOff_Common

